# $Id$
# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=popt
pkgver=1.16
pkgrel=9
pkgdesc="A commandline option parser"
arch=('i686' 'x86_64')
url="http://rpm5.org"
license=('custom')
depends=('glibc')
options=('staticlibs')
source=(http://rpm5.org/files/${pkgname}/${pkgname}-${pkgver}.tar.gz)
sha1sums=('cfe94a15a2404db85858a81ff8de27c8ff3e235e')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  ./configure --prefix=/usr
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -Dm644 COPYING "${pkgdir}"/usr/share/licenses/${pkgname}/LICENSE
}
